ME201/MTH281/ME400/CHE400
Solution of Laplace Equation
by Convolution Integral - Examples

1. Introduction

In this notebook we consider the solution of the boundary value problem given below for the Laplace equation in a two-dimensional upper half-space. The method is the Fourier transform and convolution.

$$\frac{\partial^2 \Phi}{\partial x^2} + \frac{\partial^2 \Phi}{\partial y^2} = 0\,, \quad -\infty < x < \infty,\ y > 0\,, \tag{1}$$
$$\text{with } \Phi(x, 0) = f(x), \text{ and } \Phi \underset{y\to\infty}{\longrightarrow} 0\,.$$

We obtained the solution to this problem in class by the Fourier transform, defined by

$$\tilde{\Phi}(k, y) = \int_{-\infty}^{\infty} \Phi(x, y)\, e^{-ikx}\, dx\,. \tag{2}$$

By taking the Fourier transform of the equation and boundary condition, we find the solution in the form

$$\tilde{\Phi}(k, y) = \tilde{f}(k)\, e^{-|k| y}, \tag{3}$$

where $\tilde{f}(k)$ is the Fourier transform of the boundary function $f(x)$. To invert this, we use convolution, along with the known inverse transforms

$$F^{-1}\left\{\tilde{f}(k)\right\} = f(x) \quad \text{and} \quad F^{-1}\left\{e^{-|k| y}\right\} = \frac{1}{\pi}\,\frac{y}{x^2 + y^2}. \tag{4}$$

Then the solution is

$$\Phi(x, y) = \int_{-\infty}^{\infty} \frac{1}{\pi}\,\frac{y}{(x - x')^2 + y^2}\, f(x')\, dx'\,. \tag{5}$$

2. Example

We choose a boundary potential that is constant on an interval $(-a, a)$ and zero otherwise. We let the constant value be Φ0. Then the convolution solution (using z for the variable of integration) is

In[1]:= `Clear[Φ0, a]`

In[2]:= $\Phi[x\_, y\_] = \text{Simplify}\left[\Phi0 \int_{-a}^{a} \frac{1}{\pi}\,\frac{y}{(x - z)^2 + y^2}\, dz\,, \text{Assumptions} -> \{a > 0, y > 0, x \in \text{Reals}\}\right]$

Out[2]= $\dfrac{\Phi0\left(\text{ArcTan}\left[\frac{a-x}{y}\right] + \text{ArcTan}\left[\frac{a+x}{y}\right]\right)}{\pi}$

We see that we have a very simple closed-form solution to the problem. Now we choose values for a and Φ0 and then make a contour plot of the solution. We stay slightly away from $y = 0$ to avoid problems from the y in the denominator of the arguments of the ArcTan.

```
In[3]:= a = 1.0 (** m **); Φ0 = 10.0 (** volts **);
```

```
In[4]:= ContourPlot[Φ[x, y], {x, -2 a, 2 a}, {y, 0.01, 4 a},
         FrameLabel → {"x (m)", "y (m)"}, ContourLabels → True]
```

Out[4]=



The equipotentials all reach the boundary at the discontinuities at $x = \pm a$, as we would expect. The contours look like circles. As an exercise, you might want to show analytically that they are circles.

3. A More Difficult Example

Now we look at an example in which the convolution integral has to been done numerically. We take the boundary function to be a linear function times a Gaussian:

```
In[5]:= Clear[f, Φ0, a, Φ];
```

```
In[6]:= f[x_] := Φ0 (x / a) Exp[-(x / a)^2]
```

We set the values of 100 volts for Φ0 and 1m for a.

```
In[7]:= Φ0 = 100.0; a = 1.0;
```

Before carrying out the solution, we take a look at the boundary function with the Plot command.

In[8]:=
```
Plot[f[x], {x, -3, 3}, AxesLabel → {"x (m)", "Φ (volts)"}]
```

Out[8]=



Now we do the convolution integral numerically by using *Mathematica*'s NIntegrate command. For y = 0, we use the given boundary condition for Φ.

In[9]:=
$$\Phi[x\_, y\_] := \text{If}\left[y > 0, \text{NIntegrate}\left[\frac{1}{\pi}\frac{y\, f[z]}{(x - z)^2 + y^2}, \{z, -\infty, \infty\}\right], f[x]\right]$$

Before trying a contour plot, we try a few values first.

In[10]:= `Φ[0, 1]`

Out[10]= `0.`

In[11]:= `Φ[1, 0]`

Out[11]= `36.7879`

In[12]:= `Φ[1, 1]`

Out[12]= `9.65253`

In[13]:= `Φ[-1, 1]`

Out[13]= `-9.65253`

The last two results are consistent with the antisymmetry (about x = 0) of the boundary condition.

Note that we simply specified the integration limits to be -∞ and ∞. Any numerical integral is necessarily over a finite and specific range. *Mathematica* has decided for us where to truncate the integration. That is a convenience that was not available in the old days.

Now we try the contour plot. To get a true geometric view of the contours, we choose the height and width of the plot to be the same, namely 4 m.